

Une introduction aux méthodes formelles

Guy Tremblay
Dépt. d'informatique, UQAM

Slide 1

MGL7260 Exigences et spécifications de systèmes logiciels
UQAM
31 mars 2003

Plan de la présentation :

- Qu'est-ce qu'une spécification?
- Qu'est-ce qu'une méthode formelle?
- Qu'est-ce qu'un langage formel de spécification?
- Pourquoi y-a-t-il plusieurs notations et méthodes?
- À quelles étapes les méthodes formelles peuvent-elles être utilisées?
- Quels sont les principaux bénéfices des méthodes formelles?
- Y-a-t-il des exemples d'applications développées avec des MFs?

Slide 2

Slide 3

1 Prélude : Analyse, spécification et conception

Plusieurs niveaux de description d'un système :

- (a) Les besoins et exigences des usagers ;
- (b) L'ensemble (l'espace) des solutions possibles ;
- (c) Une solution particulière ;
- (d) La structure interne de la solution choisie ;
- (e) La description des divers modules (interfaces) ;
- (f) Les algorithmes utilisés par chaque module ;
- (g) Le code final.

Slide 4

Analyse et spécification


L'étape d'**analyse et spécification** porte sur :

1. Analyse du problème = Analyser et comprendre le problème à résoudre, les besoins et exigences des usagers, les contraintes à respecter = (a)+(b)
 - (a) Besoins et exigences
 - (b) Espace des solutions possibles
2. Spécification du produit = Décrire le logiciel qui va satisfaire aux besoins et exigences et qui va respecter les contraintes = (c)
 - (c) Description d'une solution spécifique

Slide 5

Exigences et besoins vs. spécifications (selon M. Jackson) :

- Exigences et besoins = phénomènes de l'environnement
- Programmes = phénomènes internes à la machine
- Spécifications = phénomènes à l'interface entre l'environnement et la machine



Slide 6

Conception

- Conception architecturale :
 - Quels sont les principaux modules?
 - Comment sont-ils organisés et structurés?
 - Quelles sont les interfaces entre les modules?
- Conception détaillée (procédurale) :
 - Que fait précisément chacun des modules?


Analyse et spécification vs. conception

Donc, analyse et spécification vs. conception =>

- Analyse et spécification : Concepts et interfaces pour utiliser le système (point de vue de l'*usager*) ;
- Conception : Concepts et interfaces pour construire le système (point de vue du *constructeur*).

Slide 7

Pourquoi une spécification est-elle importante?



Spécification = *contrat à satisfaire*

- Validation : *Are we building the right product?* = Construit-on le bon produit?
- Vérification : *Are we building the product right?* = Construit-on le produit correctement?

Slide 8

2 Qu'est-ce qu'une "Méthode formelle"?

"[Les méthodes formelles sont des] techniques basées sur les mathématiques pour décrire des propriétés [de] systèmes [informatiques]. [Elles] fournissent un cadre [pour] spécifier, développer, et vérifier des systèmes d'une façon systématique, plutôt que d'une façon ad hoc." [Wing90]

Éléments clés d'une méthode formelle :

- Langage *formel* pour l'écriture de spécifications
- Règles pour évaluer la validité/qualité des spécifications
- Stratégies et règles pour raffiner (mettre en oeuvre) les spécifications et vérifier ces raffinements

Fondation sur laquelle *tout* repose = spécification formelle

Slide 9

3 Qu'est-ce qu'un langage *formel de spécification*?

Langage avec syntaxe et sémantique bien définies :

- Syntaxe = EBNF ; diagrammes syntaxiques ; etc.
- Sémantique = algèbres ; automates et systèmes de transitions ; fonctions, relations et prédicats ; etc.

Doit permettre de décrire le comportement d'un composant logiciel

- en décrivant ses propriétés importantes
- de façon abstraite, sans détails inutiles
- sans dire comment il est réalisé (non-algorithmique)

Un langage de programmation *n'est pas* un langage de spécification :

- trop algorithmique (opérationnelle)
- pas assez abstrait (tableaux, pointeurs, etc.)

Slide 10

Partie vraiment difficile des spécifications :

≠ mathématiques (ensembles, fonctions)

= modélisation (comprendre le problème et modéliser une solution)

Les notions mathématiques pour les spécifications basées sur l'approche de modélisation abstraite sont relativement simples :

- Logique propositionnelle :

$$p \Rightarrow q \Leftrightarrow \sim p \vee q$$
$$\sim(p \ \& \ q) \Leftrightarrow \sim p \vee \sim q$$

- Logique des prédicats :

```
SOME( i: nat :: 2*i = i )
~SOME( i: nat SUCH THAT i ~ = 0 :: 2*i = i )
ALL( i: nat :: i < i+1 )
ALL( i: nat, j: nat SUCH i < j :: i+1 < j+1 )
```

Un exemple (de M. Jackson) qui montre comment la formalisation à l'aide d'expressions logiques aide à rendre *explicite* certaines propriétés implicites.

Soit les deux affiches suivantes au pied d'un escalier mobile :

Shoes
Must Be
Worn

Dogs
Must Be
Carried

Slide 11

```
ALL( p: personne ::  
    veut_prendre_escalier(p)  
    =>  
    SOME( s: souliers :: porte_sur_ses_pieds(p, s) )  
  
ALL( p: personne, c: chien ::  
    veut_prendre_escalier_avec(p, c)  
    =>  
    porte_dans_ses_bras(p, c) )
```

- Ensembles = collections *non-ordonnées* d'éléments

```
e1: set{nat}  
e1 = {20, 30}  
add(20, e1) = e1  
size(e1 U e1) = 2  
e1 U {30..32} = {20, 30, 31, 32}  
{x: nat SUCH THAT x IN e1 :: x+1} = {21, 31}
```

Slide 12

- Séquences = suites ordonnées d'éléments :

```
s1: sequence{nat}  
s1 = [20, 30]  
length(s1) = 2  
length(s1 || s1) = 4  
add(10, s1) = [10, 20, 30]  
s1 || [30..32] = [20, 30, 30, 31, 32]
```

Slide 13

- Dictionnaires = associations entre clés et définitions :

```
d1: map{string, nat}
d1 = [{"Colin", 16}, {"Zoe", 10}; 0}
d1["Zoe"] = 10
d1["Mathieu"] = 0
domain(d1) = {"Colin", "Zoe"}
range(d1) = {0, 10, 16}
```

Spécification par opposition à implémentation

Généralement, une spécification formelle décrit ce qui doit être fait (quoi?) sans dire comment cela sera fait (comment?)

Deux exemples (en Spec) :

- Une fonction (avec plusieurs résultats acceptables) pour calculer la racine carrée d'un nombre réel :

```
r = racine@racine_carree{0.1}(4.0)
=>
1.9748417 < r < 2.0248456
```

- Une fonction pour trier une séquence de nombres :

```
s = [20, 10, 30, 20]
=>
trier@tris{nat}(s) = [10, 20, 20, 30]
```

Slide 14

Slide 15

```
FUNCTION racine_carree
  {precision: real SUCH THAT precision > 0.0}

MESSAGE racine( x: real )
  WHEN x >= 0.0
    REPLY( r: real )
      WHERE r >= 0.0 & presque_egaux( r^2, x )
  OTHERWISE
    REPLY EXCEPTION racine_imaginaire

-- Concept auxiliaire.
CONCEPT presque_egaux( v1 v2: real )
  VALUE( b: boolean )
    WHERE b <=> abs(v1 - v2) <= precision
END
```

Slide 16

```
FUNCTION tris{t: type SUCH THAT Subtype(t, total_order{t})}
  MESSAGE trier( s1: sequence{t} )
    REPLY( s2: sequence{t} )
      WHERE est_permutation(s1, s2),
            en_ordre(s2)

-- Concepts auxiliaires.
CONCEPT est_permutation( s1 s2: sequence{t} )
  VALUE( b: boolean )
    WHERE b <=> ALL(x: t :: nb_occs(x, s1) = nb_occs(x, s2))

CONCEPT nb_occs( x: t, s: sequence{t} ) VALUE( n: nat )
  WHERE n = NUMBER( i: nat SUCH THAT i IN domain(s)
                    & s[i] = x :: i )

CONCEPT en_ordre( s: sequence{t} ) VALUE( b: boolean )
  WHERE b <=> ALL( i: nat SUCH THAT 1 <= i < length(s) ::
                  s[i] <= s[i+1] )
END
```


4 Pourquoi y-a-t-il plusieurs notations et méthodes différentes

Il existe plusieurs styles différents de spécification :

- Modélisation abstraite
- Approche algébrique (types abstraits)
- Algèbre de processus
- Logique temporelle
- ...

Slide 17

Situation semblable à celle des langages de programmation :

- Domaines d'applications différents
- Styles (paradigmes) d'utilisation différents
- Pouvoirs expressifs différents

Toutes les notations n'ont pas une allure aussi *mathématique* que Z

A-1. Une machine pour une banque en Spec

```
MACHINE banque
  STATE ( soldes: map{client, montant} )
  INVARIANT ALL( c: client SUCH THAT actif(c) :: 0.0 <= soldes[c] )
  INITIALLY domain(soldes) = {}

  MESSAGE balance( c: client )
    WHEN actif(c)
      REPLY( m: montant ) WHERE m = soldes[c]
    OTHERWISE
      REPLY EXCEPTION client_invalide

  MESSAGE transfert( c: client, m: montant )
    WHEN actif(c) & (0.0 <= soldes[c] + m)
      TRANSITION
        soldes = bind( c, *soldes[c] + m, *soldes )
    WHEN actif(c) & (soldes[c] + m < 0.0)
      REPLY EXCEPTION montant_invalide
    OTHERWISE
      REPLY EXCEPTION client_invalide

  CONCEPT actif( c: client ) VALUE( b: boolean )
    WHERE b <=> c IN soldes

END
```

Slide 18

Slide 19

A-2. Une machine pour une banque en Z

Banque

$soldes: Client \rightsquigarrow Montant$

$\forall c: Client \mid actif(c, soldes) \bullet soldes(c) \geq 0.0$

InitBanque

Banque

$dom\ soldes = \{ \}$

Reponse ::=

'Ok' | 'Client invalide' | 'Montant invalide'

Slide 20

$Transfert \hat{=} TransfertOk \vee ClientInvalide \vee MontantInvalide$

TransfertOk

$\Delta Banque$

$c?: Client$

$m?: Montant$

$r!: Reponse$

$actif(c?, soldes)$

$soldes(c?) + m? \geq 0.0$

$soldes' = soldes \oplus \{c? \mapsto soldes(c?) + m?\}$

$r! = 'Ok'$

Slide 21

ClientInvalide

\exists Banque

$c?:$ Client

\neg actif($c?$, soldes)

$r!$ = 'Client invalide'

MontantInvalide

$m?:$ Montant

$r!:$ Reponse

actif($c?$, soldes)

soldes($c?$) + $m?$ < 0.0

$r!$ = 'Montant invalide'

Slide 22

B-1. Un type abstrait (collection de valeurs) pile en Spec

```
TYPE pile{T: type}
  MODEL( elems: sequence{T} )
  INVARIANT true

MESSAGE vide
  REPLY( p: pile{T} ) WHERE p.elems = []

MESSAGE estVide( p: pile{T} )
  REPLY( b: boolean ) WHERE b <=> p.elems = []

MESSAGE empiler( p: pile{T}, x: T )
  REPLY( p2: pile{T} ) WHERE p2.elems = add(x, p.elems)

MESSAGE depiler( p: pile{T} SUCH THAT ~estVide(p) )
  REPLY( p2: pile{T} )
  WHERE p2.elems = p.elems[2..length(p.elems)]

MESSAGE sommet( p: pile{T} SUCH THAT ~estVide(p) )
  REPLY( x: T ) WHERE x = p.elems[1]

END
```

Slide 23

B-2. Un type abstrait Pile en Larch (spécification algébrique)

```
Pile(T, Pile): trait
  introduces
    vide:          -> Pile
    estVide: Pile  -> Bool
    empiler: Pile, T -> Pile
    depiler: Pile  -> Pile
    sommet: Pile  -> T
  assert
    Pile generated by vide, empiler
  Forall x: T, p: Pile
    estVide(vide)          == true;
    estVide(empiler(p, x)) == false;
    sommet(empiler(p, x)) == x;
    depiler(empiler(p, x)) == p;
```

Slide 24

B-3. Une classe d'objets Pile en OCL

```
Pile
0 < taille_max
and
elems->size <= taille_max

Pile::Pile()
post: est_vide(result)

Pile::est_vide(): Boolean
post: result = elems->notEmpty

Pile::empiler( Object o )
post: elems = elems@pre->prepend(o)

Pile::depiler()
pre: elems->notEmpty
post: elems = elems@pre->subSequence(2, elems@pre->size)

Pile::sommet(): Object
pre: elems->notEmpty
post: result = elems->first
```

C. Un système réactif en Lotos et en logique temporelle

Un système est dit *réactif* lorsque ...


- il maintient une interaction *constante* avec son environnement
⇒ ne se termine jamais
- lorsque son comportement est dirigée par les événements ("*event-driven*")
⇒ des signaux/actions déclenchent des changements d'états

Description du comportement (actions et transitions) :

Slide 25

```

process MCafe[p25, p100, m25, cafe]: noexit :=
  p25; p25;
  cafe;
  MCafe[p25, p100, m25, cafe]
[]
p100;
  cafe;
  m25; m25;
  MCafe[p25, p100, m25, cafe]
endproc
  
```



Spécifications des *propriétés* du comportement :

- Automate = forme de description *opérationnelle*
≈ décrit comment générer les différentes séquences d'actions possibles
- Mais ... une telle description ne décrit pas (explicitement) les *propriétés* satisfaites par ce comportement
 - Propriétés de sûreté (*safety*): *nothing bad will ever happen.*
 - Propriétés de vivacité (*liveness*): *something good will eventually happen.*

Slide 26

Exemple = Propriétés de la machine à café exprimées en logique *modale et temporelle* :

```

not MAY_HAPPEN( <"m25"><"m25"><"m25">true )
and
not MAY_HAPPEN( <"p25"><"p25"><"cafe"><"m25">true )
and
ALWAYS( ["p100"]<"cafe"><"m25"><"m25">true )
  
```

Vérification de modèles (*model-checking*)

Model-checking = Technique *automatique* pour *vérifier* qu'un système de transitions (automate) satisfait certaines propriétés temporelles

Technique automatique \Rightarrow Un algorithme existe pour déterminer que les propriétés sont bien satisfaites

Slide 27

Exemple = Outil CADP pour vérifier les propriétés de la machine à café :

```
[MGL7260]=> caesar.adt cafe.lotos > /dev/null
[MGL7260]=> caesar cafe.lotos > /dev/null
[MGL7260]=> bcg_open cafe.bcg evaluator cafe.mcl
bcg_open: using ``/usr/local/cadp/bin.iX86/evaluator.a``
bcg_open: running ``evaluator cafe.mcl`` for ``./cafe.bcg``

TRUE
```

Autre technique de vérification = Obligations de preuve

L'application d'une *méthode* formelle (par ex., style VDM) nous oblige aussi à vérifier la qualité *intrinsèque* de cette spécification :

- Vérifier que l'état initial *établit* l'invariant
- Vérifier que chaque opération *préserve* l'invariant
- Vérifier que les réponses indiquées peuvent effectivement être produites

Cette dernière étape nécessite une connaissance du domaine d'application :

Slide 28

```
MESSAGE racine_carre( n: nat ) REPLY( r: nat )
  WHERE r^2 = n
```

Autre étape importante ... mais *informelle* = valider "cette spécification" auprès des clients

=> Retraduire en langue naturelle ou à l'aide de notations graphiques ce qui a été formalisé

Étapes ultérieures de développement = vérifier que la conception et la mise en oeuvre satisfont cette spécification

Slide 29

5 À quelles étapes les spécifications et les méthodes formelles peuvent-elles être utilisées?

- Analyse et spécification :
 - Description formelle des concepts clés du problème (domaine d'application)
 - Spécification formelle de la solution (comportement du système)
 - Prototypage
- Conception architecturale et détaillée :
 - Spécification formelle de l'interface des modules.
 - Méthode rigoureuse de raffinement
- Codification :
 - Preuves de programmes
 - Synthèse automatique du code
 - Utilisation d'assertions
- Tests
 - Génération automatique des tests

Slide 30

6 Quels sont les principaux bénéfices d'une spécification formelle?

"Devoir en arriver à une meilleure compréhension de l'élément à spécifier en forçant l'analyste à être abstrait tout en étant précis à propos des propriétés du système peut être plus gratifiant que d'avoir le document de spécification lui-même." [Wing90]

- Impact positif de l'effort de formalisation.
- Spécifications explicites, précises, non-ambiguës.
- Base pour la mise en oeuvre et pour des vérifications formelles.
- Outils (manipulation, analyse, simulation, génération de tests).
- Base pour le développement des tests.

Slide 31

7 Y-a-t-il des exemples d'applications développées à l'aide de méthodes formelles?

Exemples ayant utilisé des méthodes *classiques* :

- Réingénierie du système CICS (IBM).
- Composants matériels (Inmos).
- Appareils électroniques (Tektronik).
- Outils pour l'analyse structurée.
- Contrôle du métro (Paris).
- Contrôle de trains (SNCF).
- Contrôle du trafic aérien (UK et USA).
- Contrôle de réacteurs nucléaires (Ontario, USA).
- Navette spatiale de la NASA (contrôleur de vol, GPS).
- Contrôle pour machine de radiothérapie.

Slide 32

Exemples ayant utilisé la vérification de modèles :

- Protocole de cohérence de cache IEEE Futurebus+
(identification d'erreurs non détectées auparavant)
- Protocole de télécommunication ISDN/ISUP
(122 erreurs détectées)
- Contrôleur de canal HDLC
(erreur majeure détectée)
- Système actif de contrôle des structures en génie civil
(erreur majeure détectée, qui aurait pu *amplifier* l'effet des vibrations)
- ...

8 Conclusion

- Les méthodes formelles peuvent être utiles et peuvent jouer un rôle important dans le développement de systèmes critiques
- Peuvent apparaître coûteuses à première vue (nouvelle approche) mais peuvent avoir pour effet de réduire les coûts des phases subséquentes (mise en oeuvre, tests)
- ... mais ne sont pas la panacée (le *silver bullet*) = il faut savoir les utiliser à bon escient (*lightweight formal methods*)

Slide 33

- Pour en savoir plus :
 - Cours MGL7160 "Méthodes formelles et semi-formelles".
 - *Modélisation et spécification formelle des logiciels*, G. Tremblay, Loze-Dion Éditeurs, 2000.
 - *The Object Constraint Language — Precise Modeling with UML*, J. Warmer and A. Kleppe, Addison-Wesley, 1999.
 - *Model Checking*, E. Clarke, O. Grumberg et D.A. Peled, The MIT Press, 1999.